<commit_before>{ stdenv, fetchurl, m4, perl, lzma }:

stdenv.mkDerivation (rec {
  name = "libtool-2.4.2";

  src = fetchurl {
    url = "mirror://gnu/libtool/${name}.tar.gz";
    sha256 = "0649qfpzkswgcj9vqkkr9rn4nlcx80faxpyqscy2k1x9c94f93dk";
  };

  nativeBuildInputs = [ lzma m4 perl ];

  # Don't fixup "#! /bin/sh" in Libtool, otherwise it will use the
  # "fixed" path in generated files!
  dontPatchShebangs = true;

  # XXX: The GNU ld wrapper does all sorts of nasty things wrt. RPATH, which
  # leads to the failure of a number of tests.
  doCheck = false;

  meta = {
    description = "GNU Libtool, a generic library support script";

    longDescription = ''
      GNU libtool is a generic library support script.  Libtool hides
      the complexity of using shared libraries behind a consistent,
      portable interface.

      To use libtool, add the new generic library building commands to
      your Makefile, Makefile.in, or Makefile.am.  See the
      documentation for details.
    '';

    homepage = http://www.gnu.org/software/libtool/;

    license = stdenv.lib.licenses.gpl2Plus;

    maintainers = [ ];
  };
}

//

# Don't run the native `strip' when cross-compiling.  This breaks at least
# with `.a' files for MinGW.
(if (stdenv ? cross)
 then { dontStrip = true; }
 else { }))
<commit_msg>libtool: update<commit_after>{ stdenv, fetchurl, m4, perl, help2man }:

stdenv.mkDerivation rec {
  name = "libtool-2.4.5";

  src = fetchurl {
    url = "mirror://gnu/libtool/${name}.tar.gz";
    sha256 = "11v9gl8ancavx7179n6zk4k8hwa0zl4wz4w9z3mffk71gnfb972h";
  };

  propagatedNativeBuildInputs = [ m4 ];
  nativeBuildInputs = [ perl help2man ];

  # Don't fixup "#! /bin/sh" in Libtool, otherwise it will use the
  # "fixed" path in generated files!
  dontPatchShebangs = true;

  # XXX: The GNU ld wrapper does all sorts of nasty things wrt. RPATH, which
  # leads to the failure of a number of tests.
  doCheck = false;

  # Don't run the native `strip' when cross-compiling.  This breaks at least
  # with `.a' files for MinGW.
  dontStrip = stdenv ? cross;

  meta = {
    description = "GNU Libtool, a generic library support script";

    longDescription = ''
      GNU libtool is a generic library support script.  Libtool hides
      the complexity of using shared libraries behind a consistent,
      portable interface.

      To use libtool, add the new generic library building commands to
      your Makefile, Makefile.in, or Makefile.am.  See the
      documentation for details.
    '';

    homepage = http://www.gnu.org/software/libtool/;

    license = stdenv.lib.licenses.gpl2Plus;

    maintainers = [ ];
  };
}


{'fraction_non_alphanumeric': 0.1019563581640331, 'fraction_numerical': 0.02144469525959368, 'mean_word_length': 3.0471841704718416, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7242271', 'n_tokens_mistral': 927, 'n_tokens_neox': 885, 'n_words': 282}